

86027



U.S. Department
of Transportation
**National Highway
Traffic Safety
Administration**

Design and Development of an Advanced ATD Thorax System for Frontal Crash Environments

Final Report

Volume 3:
User Guide for Chest Deflection
Measurement System

Trauma Assessment Device Development Program

1. Report No.		2. Government Accession No. DOT HS 808 140		3. Recipient's Catalog No.	
4. Title and Subtitle DESIGN AND DEVELOPMENT OF AN ADVANCED ATD THORAX SYSTEM FOR FRONTAL CRASH ENVIRONMENTS, VOLUME 3: USER GUIDE FOR CHEST DEFLECTION MEASUREMENT SYSTEM				5. Report Date June 1992	
				6. Performing Organization Code	
7. Authors Lawrence W. Schneider, Bruce M. Bowman, Richard J. Lehman				8. Performing Organization Report No. UMTRI-92-22-3	
9. Performing Organization Name and Address University of Michigan Transportation Research Institute 2901 Baxter Road Ann Arbor, Michigan 48109				10. Work Unit No.	
				11. Contract or Grant No. DTNH22-83-C-07005	
12. Sponsoring Agency Name and Address U.S. Department of Transportation National Highway Traffic Safety Administration 400 Seventh Street S.W. Washington, D.C. 20590				13. Type of Report and Period Covered Final Report	
				14. Sponsoring Agency Code	
15. Supplementary Notes Mark P. Haffner, NHTSA COTR					
16. Abstract The Prototype-50M assembly described in Volume 1 incorporates new ribcage geometry and chest deflection instrumentation capable of measuring three-dimensional displacements at four critical injury locations on the ribcage. This volume describes the double-gimballed string potentiometer (DGSP) transducer units used to measure ribcage displacements and defines the coordinate systems in which these displacements are described. A FORTRAN computer subprogram called DEFLECT was developed to compute the displacements from the initial positions and output of the DGSP units. Procedures for using this program are described and a listing of the FORTRAN code is provided at the end of the document.					
17. Key Words Chest Deflection, Crash Dummy Instrumentation			18. Distribution Statement Document is available to the public from the National Technical Information Service, Springfield, Virginia 22161		
19. Security Classif. (of this report) NONE		20. Security Classif. (of this page) NONE		21. No. of Pages 32	22. Price

This publication is distributed by the U.S. Department of Transportation, National Highway Traffic Safety Administration, in the interest of information exchange. The opinions, findings, and conclusions expressed in this publication are those of the author(s) and not necessarily those of the Department of Transportation or the National Highway Traffic Safety Administration. The United States Government assumes no liability for its contents or use thereof. If trade or manufacturers' names or products are mentioned, it is only because they are considered essential to the object of the publication and should not be construed as an endorsement. The United States Government does not endorse products or manufacturers.

ACKNOWLEDGEMENTS

The authors wish to acknowledge the contributions of Michael J. Salloum of First Technology Safety Systems, Inc., Mark P. Haffner of the NHTSA, and Brian J. Eby of UMTRI, who, together with Lawrence W. Schneider, designed and developed the double-gimballed string potentiometer (DGSP) transducers.

Special recognition is also due to Warren Hardy of Wayne State University, who developed the calibration test facility and performed calibration tests to evaluate the accuracy and durability of these devices; to Alena Hagedorn of Transportation Research Center, Inc., who helped debug the DEFLECT subprogram used to calculate three-dimensional displacements from the DGSP transducers; and to John Gates of Space Age Controls, Inc., who served as a resource and provided support to the instrumentation design effort.

CONTENTS

ACKNOWLEDGEMENTS	iii
LIST OF TABLES	vii
LIST OF FIGURES	ix
1. DESCRIPTION OF MEASUREMENT SYSTEM AND DGSP TRANSDUCERS	1
2. COMPRESSION IN SPINAL AXIS SYSTEM	6
3. COMPRESSION IN ALTERNATE AXIS SYSTEM	6
4. INITIAL TRANSDUCER ORIENTATIONS	7
5. SUBROUTINE "DEFLECT"	10
6. SOURCE CODE FOR SUBROUTINE "DEFLECT"	15

LIST OF TABLES

	Page
1. Projected Angles of Alternate Compression Axes Relative to Spinal Axis System	6
2. Initial Projected Angles of Joy Sticks in Spinal Axis System	7
3. Variables for Input Scalars from Calling Program to Subroutine DEFLECT	10
4. Variables for Input Arrays from Calling Program to Subroutine DEFLECT	11
5. Variables for Returned Arrays from Subroutine DEFLECT to Calling Program	11

LIST OF FIGURES

		Page
1.	Prototype-50M showing locations of DGSP units on spine and ribcage	2
2.	Side-view drawing of Prototype-50M thorax assembly showing relationship of ribcage, sternum, and DGSP transducer mounting locations on lower thoracic spine	3
3.	A double-gimballed string potentiometer (DGSP) with telescoping joy stick prior to installation in the lower thoracic spine segment	4
4.	Directions of axes in lower thoracic spine coordinate system	4
5.	Illustration of DGSP mounted to lower thoracic spine segment showing directions of gimbal rotations	5
6.	Schematic drawing of double-gimballed joy stick (DGSP)	5
7.	Side-view drawing of Prototype-50M thorax showing directions of compression axes at the sternum and lower ribcage	8
8.	Top-view sketch showing orientations of compression axes at the lower ribcage	8
9.	Initial orientations of DGSPs specified by ϕ_{P_n} and θ_{P_n}	9

USER GUIDE FOR CHEST DEFLECTION MEASUREMENT SYSTEM

1. DESCRIPTION OF MEASUREMENT SYSTEM AND DGSP TRANSDUCERS

The Prototype-50M thorax shown and illustrated in Figures 1 and 2 is equipped with new instrumentation that provides for three-dimensional displacement measurement at four points on the new ribcage. Two of these points are located at the anterior ends of the third rib to the left and right of the sternum, corresponding approximately to the level of the fourth/fifth rib interspace in the human. This is also the level of the center of the impactor in the Kroell et al. (1974)¹ pendulum tests to cadaver chests. The other two points are located in the regions of the left and right lower ribcage at the ends of the Prototype-50M rib six (i.e., the next-to-bottom rib). These points correspond to the level of the eighth rib in the human approximately over the liver and spleen areas.

Each of the four transducer units consists of a string potentiometer mounted in a double-gimballed mechanism (a double-gimballed string potentiometer or DGSP) that mounts to the lower thoracic spine segment. Figure 3 shows one DGSP unit prior to installation in the lower thoracic spine. A coordinate system for this lower thoracic spine segment is defined as shown in Figure 4 where the Z-axis is in the midsagittal plane in the direction of the long axis of the spine segment, the X-axis is in the midsagittal plane perpendicular to the Z-axis, and the Y-axis is perpendicular to the X-Z plane. The positive directions are chosen to define a right-hand coordinate system so that X is positive toward the front, Y is positive toward the left, and Z is positive up. The location of the coordinate system origin is not important since chest displacements are changes from the initial position and not absolute values.

As shown in Figure 5, for each transducer unit, one gimbal is mounted in the side of the spine so that the axis of rotation is fixed in the lateral or Y direction of the spinal coordinate system. Rotation about this axis is referred to by the symbol ψ (ψ). The other gimbal is mounted in the yoke bracket surrounding the string potentiometer and the shaft of the yoke rotates in the first or spine gimbal about the spinal Y-axis. The axis of rotation for this second gimbal, therefore, remains perpendicular to the axis of the first gimbal but rotates in the spinal X-Z plane with rotation of the first gimbal about the Y-axis in the spine. Rotation about this gimbal axis is referred to by the symbol θ (θ). It should be noted that because the meaning of rotation in θ regarding displacement in a dummy coordinate system is altered by rotation in ψ , it is necessary to use Euler-angle terminology and analysis when computing chest displacements from this transducer system.

As illustrated in Figure 6, the cable of each string potentiometer is threaded around an idler pulley that is installed in the modified string potentiometer housing. The vector formed by the cable goes through the center of the string potentiometer pulley and the coincident centers of the two gimbal axes. This cable is threaded through, and attached to, the end of a telescoping joy stick, which is attached to the string potentiometer housing at its base and connected to the rib end by means of a universal joint.

The joy stick provides the cable with lateral stiffness so that angular motions of the gimbals will follow motions of the rib ends that are not in the direction of the cable vector (i.e., motions other than inward and outward displacements of the cable). Rotary potentiometers mounted in line with the gimbal axes provide electrical signals corresponding

¹Kroell, C.K.; Schneider, D.C.; and Nahum, A.M. (1974) Impact tolerance and response to the human thorax II. *Proc. 18th Stapp Car Crash Conference*, pp. 383-457. Society of Automotive Engineers, Warrendale, Pa.




FIGURE 1. Prototype-50M showing locations of DGSP units on spine and ribcage.




FIGURE 2. Side-view drawing of Prototype-50M thorax assembly showing relationship of ribcage, sternum, and DGSP transducer mounting locations on lower thoracic spine.




FIGURE 3. A double-gimbal string potentiometer (DGSP) with telescoping joy stick prior to installation in the lower thoracic spine segment.




FIGURE 4. Directions of axes in lower thoracic spine coordinate system.




FIGURE 5. Illustration of DGSP mounted to lower thoracic spine segment showing directions of gimbal rotations.




FIGURE 6. Schematic drawing of double-gimbal joy stick (DGSP).

to the angular movements of the telescoping joy stick from its initial position. The initial orientation of each string potentiometer cable relative to the spinal axis system and the initial distance, R_0 , from the center of each pulley to the end of each joy stick are used along with the change in cable length and changes in angles of the gimbals to compute the three-dimensional displacements at each of the four points on the ribcage. As will be described, these displacements are computed along the spinal axes and with respect to any other user-defined compression axis system in the Prototype-50M chest.

2. COMPRESSION IN SPINAL AXIS SYSTEM

The computer subprogram DEFLECT uses the three output signals of each of the four DGSPs in the Prototype-50M chest to compute the three-dimensional displacements (i.e., ΔX , ΔY , and ΔZ) of the ribcage at four points. As will be described in greater detail below, the output of the program provides these displacements in two coordinate systems. The program computes the anterior-posterior (AP), lateral (RL), and inferior-superior (IS) displacements of the four ribcage points in terms of a lower thoracic spine coordinate system defined above and shown in Figure 4, where IS displacement is along the Z-axis, AP displacement or chest compression is along the X-axis, and RL displacement is along the Y-axis. When the Prototype-50M is positioned for pendulum calibration testing with the surface of the sternum approximately vertical, the spinal AP axis is approximately in the direction of the impact (see Figure 4) and corresponds closely to the direction in which chest deflection was measured in the Kroell et al. tests.

3. COMPRESSION IN ALTERNATE AXIS SYSTEM

In addition, the program computes three-dimensional displacements in a second coordinate system defined by the user for each transducer location. In this system, the X-axis defines the direction in which chest compression is to be measured. The alternate orientations of these alternate compression axes for the four transducer sites used in dummy development are taken to be approximately perpendicular to the anterior surface of the ribcage at each of the four measurement sites and are summarized in Table 1. The angles given are the planar *projected* angles of each of the compression axes relative to the spinal X-axis. Using the right-hand rule and the definition of the spinal axis system as a right-hand coordinate system, the signs of these projected angles, α and β , are defined as:

- leftward rotation about the Z-axis is positive (inward for right-side rod),
- downward rotation about the Y-axis is positive.

TABLE 1
PROJECTED ANGLES OF ALTERNATE COMPRESSION AXES
RELATIVE TO SPINAL AXIS SYSTEM
(All angles given in degrees.)

Transducer No. (n)	Transducer Location	Projected angle onto X-Z plane of spine axis system (Rotation about spinal Y-axis.) (β_n)	Projected angle onto X-Y plane of spine axis system (Rotation about spinal Z-axis.) (α_n)
1	Right sternum	0	0
2	Left sternum	0	0
3	Right lower ribcage	+15 (down)	-18 (to right)
4	Left lower ribcage	+15 (down)	+18 (to left)

*NOTE that the left and right are defined as the dummy's left and right.

As indicated by the zero values for α and β in the top two rows, the alternate compression axes at the sternum (i.e., at rib 3) are currently defined in the same direction as the spinal X-axis as shown in Figure 7. If, however, the user wishes to define a different compressive axis direction in this region so that, for example, the sternal compression axes are inclined 10 degrees to the spinal X-axis, the values of the appropriate projected angles (α , β) can be changed in the user's program before calling the subroutine DEFLECT (see Section 5).

For the two transducer units attached to the lower ribcage, the orientations of the alternate compression axes are currently defined so that the directions of chest compression are outward 18 degrees and downward 15 degrees from the spinal X-axis, as illustrated in Figures 7 and 8. These definitions of chest compression can also be changed, however, by changing the appropriate variable values in the calling program. It is emphasized that the orientations of these compression axes are defined relative to the spinal X-axis by angles that represent the planar projections of the compression axes onto the spinal axis system.

4. INITIAL TRANSDUCER ORIENTATIONS

In addition to specifying the projected angles of the alternate compression axes for each of the transducer units, the user must also provide information specifying the initial positions and orientations of the string potentiometer cables relative to the spinal axis system. This includes the initial cable length for each transducer unit, R_0 , defined as the distance from the center of the string potentiometer and gimbal axes to the center of the universal joint at the rib interface, as well as the initial projected angles of each joy stick with the spinal axis system. The latter include both the projected angle onto the spinal X-Z plane (i.e., rotation about the spinal Y-axis), indicated by ψ_{P_n} in Table 2 and Figure 9, as well as the projected angle onto the X-Y plane (i.e., rotation about the spinal Z-axis), indicated by θ_{P_n} in Table 2 and Figure 9. Since the algorithm for computing chest displacements requires that the outputs of the gimbal transducers be considered as Euler angles, where the interpretation of the Z-axis gimbal is dependent on the rotation of the Y-axis gimbal in the spine, the computer program computes the appropriate Euler angle initial conditions from the projected angles provided by the user.

The initial projected angles defining the orientations of the four transducer units in the Prototype-50M chest are currently as given in Table 2 and illustrated in Figure 9. Again, the signs of the projected angles are determined by the definition of a right-hand spinal coordinate system and the right-hand rule so that the projected angles with respect to the spinal X-axis are positive for rotation to the left about the Z-axis (θ_{P_n}) and rotation downward about the Y-axis (ψ_{P_n}).

TABLE 2

INITIAL PROJECTED ANGLES OF JOY STICKS IN SPINAL AXIS SYSTEM
(All angles given in degrees.)

Transducer No. (n)	Transducer Location	Projected angle onto X-Z plane of spine axis system (Rotation about spinal Y-axis.) (ψ_{P_n})	Projected angle onto X-Y plane of spine axis system (Rotation about spinal Z-axis.) (θ_{P_n})
1	Right sternum	0	+10 (in)
2	Left sternum	0	-10 (in)
3	Right lower ribcage	+6 (down)	-5 (out)
4	Left lower ribcage	+6 (down)	+5 (out)

*NOTE that the left and right are defined as the dummy's left and right.




FIGURE 7. Side-view drawing of Prototype-50M thorax showing directions of compression axes at the sternum and lower ribcage. Y-axis is positive out of the page.




FIGURE 8. Top-view sketch showing orientations of compression axes at the lower ribcage.





FIGURE 9. Initial orientations of DGSPs specified by ϕ_{P_n} and θ_{P_n} .

5. SUBROUTINE "DEFLECT"

The subroutine DEFLECT is a FORTRAN subprogram that performs the calculations involved in generating three-dimensional displacement information from the length, and angular data produced by the double-gimballed string potentiometer transducers (DGSP). It was compiled and tested using Lahey FORTRAN F77L (version 5.0). The only compiler-specific features in the program are the in-line comments (indicated by an exclamation point, or !), which may be removed without affecting the program's operation.

The routine makes no assumptions about the initial conditions of the DGSP. Initial angles and joy-stick length must be provided by the user's calling program as described previously. Joy-stick movement is computed for one DGSP unit (i.e., unit 1, 2, 3, or 4) each time the subroutine is called, so that for a complete analysis of the four DGSPs in the Prototype-50M chest, the routine must be called four times with appropriate initial conditions and compressive axes orientations given in each call. It is also expected that all signal data sent to the subroutine will have been scaled and filtered appropriately, that any zero-offset bias will have been removed, and that signs for the different signals are correct according to the conventions given previously for the spinal right-hand coordinate system.

The subroutine returns the chest deflection information to the calling program in the spinal compression axis system (i.e., where the -X axis of the spine is the compressive axis) and in the user-defined or alternate compression axis system. The former are contained in the DELX(i), DELY(i), and DELZ(i) arrays, while the latter are in the STRETCH(i), SWING_L(i), and LIFT(i) arrays. In addition, the X(i), Y(i), and Z(i) arrays contain the coordinates of the end of the joy stick in the spinal axis system at each point in time.

Tables 3 through 5 summarize the calling argument scalars and arrays and the returned arrays.

TABLE 3
VARIABLES FOR INPUT SCALARS FROM CALLING PROGRAM
TO SUBROUTINE "DEFLECT"

Item	Type	Description of Contents
NPTS	INT*4	Number of points in arrays DR, DTHETA, etc.
RO	REAL*4	Initial length of DGSP probe, mm.
THETAP	REAL*4	Angle of joy stick in its initial orientation made by projecting onto the X-Y plane, degrees. (Positive CCW when viewed from above.)
PSIP	REAL*4	Angle of joy stick in its initial orientation made by projecting onto the X-Z plane, degrees. (Positive CCW when viewed from left.)
ALPHA	REAL*4	Projected angle of alternate compression vector onto X-Y plane of spinal coordinate system axis, degrees. (Positive CCW when viewed from above.)
BETA	REAL*4	Projected angle for alternate compression vector onto X-Z plane, degrees. (Positive CCW when viewed from left.)

TABLE 4
VARIABLES FOR INPUT ARRAYS FROM CALLING PROGRAM
TO SUBROUTINE "DEFLECT"

Item	Type	Description of Contents
DR(npts)	REAL*4	Deflection of transducer probe, mm. (Positive in +X or forward direction.)
DTHETA(npts)	REAL*4	Change of angle of rotation around Z-axis gimbal, deg. (Positive CCW when viewed from above.)
DPSI(npts)	REAL*4	Change of angle of rotation around Y-axis gimbal, deg. (Positive CCW when viewed from left.)

TABLE 5
VARIABLES FOR RETURNED ARRAYS FROM
SUBROUTINE "DEFLECT" TO CALLING PROGRAM

Item	Type	Description of Contents
X(npts), Y(npts), Z(npts)	REAL*4	Computed (x,y,z) position of end of probe in spine coordinates.
DELX(npts), DELY(npts), DELZ(npts)	REAL*4	Computed change in position of end of probe in spine coordinates. Note: DELX[i]←X[i] - X[1]
STRETCH(npts)	REAL*4	Computed elongation (negative compression). (Alternate axis system.)
SWING_L(npts)	REAL*4	Computed leftward shift. (Alternate axis system.)
LIFT(npts)	REAL*4	Computed upward lift. (Alternate axis system.)

USER GUIDE

Since the ALPHAs, BETAs, THETAPs, and PSIPs are projected angles, they must be converted to their respective Euler angles, THETA and PSI, and all angles must be converted from degrees to radians for computational purposes.

The first step is the conversion of the projected angles PSIP and THETAP to Euler angles by the following equations where degrad is the factor for converting degrees to radians.

$$\begin{aligned}\text{PSI0} &= \text{PSIP} * \text{degrad} \\ \text{THETA0} &= \text{atan}(\tan(\text{THETAP} * \text{degrad}) * \cos(\text{PSIP} * \text{degrad}))\end{aligned}$$

The initial positions (indicated by a variable name suffixed with `_init`) are the input values adjusted by the amount of the first digitized datum:

$$\begin{aligned}\text{R_init} &= \text{R0} - \text{DR}(1) \\ \text{PSI_init} &= \text{PSI0} - \text{DPSI}(1) * \text{degrad} \\ \text{THETA_init} &= \text{THETA0} - \text{DTHETA}(1) * \text{degrad}\end{aligned}$$

Similarly, the projected angles of the alternate axis system are converted to Euler angles. Some constants which will be used later in the loop are calculated at the same time.

$$\begin{aligned}\text{PSI_AB} &= \text{beta} * \text{degrad} \\ \text{THETA_AB} &= \text{atan}(\tan(\text{alpha} * \text{degrad}) * \cos(\text{beta} * \text{degrad})) \\ \text{cos_PSI_AB} &= \cos(\text{PSI_AB}) \\ \text{cos_THETA_AB} &= \cos(\text{THETA_AB}) \\ \text{sin_PSI_AB} &= \sin(\text{PSI_AB}) \\ \text{sin_THETA_AB} &= \sin(\text{THETA_AB})\end{aligned}$$

Initial probe endpoints are calculated using the following equations:

$$\begin{aligned}\text{X1} &= \text{R_init} * \cos(\text{PSI_init}) * \cos(\text{THETA_init}) \\ \text{Y1} &= \text{R_init} * \sin(\text{THETA_init}) \\ \text{Z1} &= -\text{R_init} * \cos(\text{THETA_init}) * \sin(\text{PSI_init})\end{aligned}$$

Then new endpoints and deflections are computed for each time point. First the change in probe length and angle are obtained:

$$\begin{aligned}\text{R} &= \text{R_init} + \text{DR}(I) \\ \text{THETA} &= \text{THETA_init} + \text{DTHETA}(I) * \text{degrad} \\ \text{PSI} &= \text{PSI_init} + \text{DPSI}(I) * \text{degrad}\end{aligned}$$

Then, new endpoints and deflections in three axes are computed:

$$\begin{aligned}\text{cos_THETA} &= \text{COS}(\text{THETA}) \\ \text{X}(i) &= \text{R} * \text{COS}(\text{PSI}) * \text{cos_THETA} \\ \text{Y}(i) &= \text{R} * \text{SIN}(\text{THETA}) \\ \text{Z}(i) &= -\text{R} * \text{SIN}(\text{PSI}) * \text{cos_THETA} \\ \text{DELX}(i) &= \text{X}(i) - \text{X1} \\ \text{DELY}(i) &= \text{Y}(i) - \text{Y1} \\ \text{DELZ}(i) &= \text{Z}(i) - \text{Z1}\end{aligned}$$

Finally, a set of deflections for the (optional) alternate axis system are derived:

$$\begin{aligned}\text{STRETCH}(i) &= \text{DELX}(i) * \cos_THETA_AB * \cos_PSI_AB \\ &\quad + \text{DELY}(i) * \sin_THETA_AB \\ &\quad - \text{DELZ}(i) * \cos_THETA_AB * \sin_PSI_AB \\ \text{SWING_L}(i) &= -\text{DELX}(i) * \sin_THETA_AB * \cos_PSI_AB \\ &\quad + \text{DELY}(i) * \cos_THETA_AB \\ &\quad + \text{DELZ}(i) * \sin_THETA_AB * \sin_PSI_AB \\ \text{LIFT}(i) &= \text{DELX}(i) * \sin_PSI_AB \\ &\quad + \text{DELZ}(i) * \cos_PSI_AB\end{aligned}$$

(Note that if ALPHA = BETA = 0.0, then STRETCH = DELX, SWING_L = DELY and LIFT = DELZ.)

These calculations are carried out for each time point. The results are returned to the calling program in the arrays described above.

6. SOURCE CODE FOR SUBROUTINE "DEFLECT"


```

=====
c
c Subroutine Name: DEFLECT
c
c Purpose:
c       To calculate three-dimensional displacements of
c       the Prototype-50M (TAD-50M) ribcage using output
c       of the double-gimballed string potentiometers
c
c Measurement Concept:
c
c       L.W. Schneider, Research Scientist
c       Head, Biosciences Division, UMTRI
c
c       M.P. Haffner, Contracting Officer's Technical Representative
c       Biomechanics Division, Office of Crashworthiness Research
c       U.S. Department of Transportation
c       National Highway Traffic Safety Administration
c
c Mathematical Analyses:
c
c       B.M. Bowman, Research Scientist,
c       Biosciences Division, UMTRI
c
c FORTRAN coding:
c
c       R.J. Lehman, Senior Program Analyst,
c       Biosciences Division, UMTRI
c
c Release Date:
c
c       31 July 1992
c
c The authors would like to thank Joseph A. Prater for his assistance
c in the development of this subroutine.
c
=====
c
c       subroutine deflect(npts, r0,           ! input scalars and
1      ThetaP, PsiP, alpha, beta,           ! projection angles,
1      dr, dtheta, dpsi,                    ! input vectors
2      x, y, z, delx, dely, delz,           ! returned spine coords
3      stretch, swing_L, lift)             ! returned "other" coords
c
c       real*4      dr(1), dtheta(1), dpsi(1),
1      x(1), y(1), z(1),
2      delx(1), dely(1), delz(1),
3      stretch(1), swing_L(1), lift(1),
4      r0, ThetaP, PsiP, alpha, beta
c       integer*4   npts
c       real*4      degrad
c
-----
c
c Definitions:
c
c Axes are:  X --> forward  [Posterior  >>  Anterior]
c            Y --> left,   [Right      >>  Left]
c            Z --> up     [Inferior   >>  Superior]
c
c Rotations: Psi:  Rotation around Y-axis gimbal,
c                 (Positive CCW when viewed from left.)
c                 Theta: Rotation around Z-axis gimbal,
c                 (Positive CCW when viewed from above.)
c

```

USER GUIDE

```

c Calling arguments:
c
c   Initial conditions (scalars):
c
c   Name           Type      Contents-----
c
c   R0              R*4      Rod initial length (input)
c   ThetaP          R*4      Initial value of angle made with X-axis by
c                          projection of rod onto X-Y plane. (Positive
c                          from X toward Y; "outward" for left side rod,
c                          "inward" for right side rod.)
c   PsiP            R*4      Initial value of angle made with X-axis by
c                          projection of rod onto X-Z plane. (Positive
c [5/11/92]          from X toward -Z, i.e., "DOWN".)
c   ALPHA           R*4      Projection angle for alternate deflection axis,
c [5/14/92]          positive for X --> Y (deg)
c   BETA            R*4      Projection angle for alternate deflection axis,
c [5/14/92]          positive for X --> -Z (deg)
c
c   NPTS            I*4      number of points in arrays DR, DTHETA, etc.
c
c   Sensor values, zeroed and scaled (vectors):
c
c   DR(npts)        R*4      Change of deflection of transducer rod/string
c                          (Positive in +X [forward] direction)
c   DTHETA(npts)    R*4      Change of Angle of rotation around Z-axis gimbal,
c                          degrees (Positive CCW when viewed from above.)
c   DPSI(npts)      R*4      Change of Angle of rotation around Y-axis gimbal,
c                          degrees (Positive CCW when viewed from left.)
c
c   Returned Spinal Coordinate Values:
c
c   X(npts),Y(),Z() R*4      Computed (x,y,z) position of end of rod in
c                          spine coordinates
c
c   DELX(npts),     R*4      Computed change in A-P position of end of rod;
c                          DELX[i] <--- X[i] - X[1], etc.
c   DELY(npts),     R*4      Computed change in R-L position of end of rod;
c   DELZ(npts),     R*4      Computed change in I-S position of end of rod;
c
c   Returned Alternate or Compression Coordinate Values:
c
c   STRETCH(npts)   R*4      Computed stretch
c   swing_L(npts)   R*4      Computed leftward shift
c   LIFT(npts)      R*4      Computed upward lift
c
c-----
c
c   degrad = 4. * atan(1.0) / 180.      ! degrees --> radians factor
c
c   convert projected angles to radians; minor name change to prevent
c   altering calling argument
c
c   ThPRad = ThetaP * degrad
c   alfa = alpha * degrad
c   PsPRad = PsiP * degrad
c   bet = beta * degrad
c
c   calculate initial positions
c   (dr(1), dtheta(1), dpsi(1) are approx = 0.0,
c   so r_init, theta_init, psi_init are very close to initial values.
c
c   r_init = R0 - dr(1)                ! changed from (R0+dr(1)) 5/11/92
c
c   compute Euler angles given projected angles:
c

```

```

psi0 = PsPRad                                ! changed from (-PsPRad) 5/11/92
psi_init = psi0 - dpsi(1) * degrad

theta0 = atan(tan(ThPRad)*cos(PsPRad))
theta_init = theta0 - dtheta(1) * degrad

c
c   similarly for offset angles
c
psi_AB = bet                                  ! changed from (-bet) 5/11/92
theta_AB = atan(tan(alfa)*cos(bet))
cos_PSI_AB = cos(psi_AB)
cos_THETA_AB = cos(theta_AB)
sin_PSI_AB = sin(psi_AB)
sin_THETA_AB = sin(theta_AB)

c
c x(i), y(i), z(i) are endpoints of probe at time i. If i=1, initial condition.
c

x1 = r_init * cos(psi_init) * cos(theta_init) ! X initial position
y1 = r_init * sin(theta_init)                ! Y initial position
z1 = -r_init * cos(theta_init) * sin(psi_init) ! Z initial position

c
c compute new endpoints and deflection for each time point
c
do 20 i=1,npts
c
c R, THETA, PSI are rod length and angle
c value from xducer at time i plus initial condition.
c
R = R_init + DR(I)
THETA = THETA_init + DTHETA(I) * degrad
PSI = PSI_init + DPSI(I) * degrad

c
c so new endpoints (in spine reference frame) are:
c
X(i) = R * COS(PSI) * COS(THETA)
Y(i) = R * SIN(THETA)
Z(i) = -R * SIN(PSI) * COS(THETA)

c
c and deflection components are:
c
DELX(i) = X(i) - X1
DELY(i) = Y(i) - Y1
DELZ(i) = Z(i) - Z1

c
c and for second reference frame, (from bmb, 5/14/92)
c
stretch(i) =      DELX(i) * cos_THETA_AB * cos_PSI_AB
1              + DELY(i) * sin_THETA_AB
2              - DELZ(i) * cos_THETA_AB * sin_PSI_AB

swing_L(i) =      - DELX(i) * sin_THETA_AB * cos_PSI_AB
1              + DELY(i) * cos_THETA_AB
2              + DELZ(i) * sin_THETA_AB * sin_PSI_AB

lift(i) =         DELX(i) * sin_PSI_AB
1              + DELZ(i) * cos_PSI_AB
20 continue

c
c and we're done.
c
return
end

```

USER GUIDE

```
=====
c Subroutine Name:  SETDEFL
c
c Purpose:  To Initialize joystick angles and lengths and specify the
c           projected angles of alternate compression axes to be used at
c           the sternum and lower ribcage.  This subprogram is called by
c           the user's main program prior to calling the subroutine DEFLECT.
c           The user may wish to use alternative approaches to initializing
c           these variables.
c
c Author:    R.J. Lehman
c
c Release Date:  31 July 1992
c
=====
c
c      subroutine setdefl
c      real*4      r0,      ThetaP,      PsiP,      alpha,      beta
c      common /inits/r0(4), ThetaP(4), PsiP(4), alpha(4), beta(4)
c
c      common block /inits/      contains the initial, measured lengths and
c                               angular orientation of the transducers.
c                               Assignments are as given in XDUCER (below)
c
c      ***** Note: ***** >>      These values must be supplied by the user
c
c      r0(4)      R*4      Initial length of probe or string.
c      ThetaP(4) R*4      initial rod projection angle in spine system, X --> Y
c      PsiP(4)   R*4      initial rod projection angle in spine system, X --> -Z
c      ALPHA(4)  R*4      projection angle for alternate deflection axis, X --> Y (deg)
c      BETA(4)   R*4      projection angle for alternate deflection axis, X -->-Z (deg)
c
=====
c      Transducer Identification:
c
c                               1 = right upper
c                               2 = left upper
c                               3 = right lower
c                               4 = left lower
c
c
c      r0(1) = 152.4
c      r0(2) = 152.4
c      r0(3) = 149.86
c      r0(4) = 149.86
c
c      ThetaP(1) = 10.
c      ThetaP(2) = -10.
c      ThetaP(3) = -5.
c      ThetaP(4) = 5.
c
c      PsiP(1) = 0.
c      PsiP(2) = 0.
c      PsiP(3) = 6.
c      PsiP(4) = 6.
c
c      alpha(1) = 0.
c      alpha(2) = 0.
c      alpha(3) = -18.
c      alpha(4) = 18.
c
c      beta(1) = 0.
c      beta(2) = 0.
c      beta(3) = 15.
c      beta(4) = 15.
c
c      return
c      end
```